```
bin[z_, k_] := Product[z - j, {j, 0, k - 1}] / k!
p[n_] := p[n] = If[PrimeQ[n], 1, 0]
lp[n_, 0] := UnitStep[n - 1]
lp[n_, k_] := lp[n, k] = Sum[ If[p[j] == 0, 0, lp[Floor[n / j], k - 1]], {j, 2, n}]
lz[n_, z_] := Sum[ z^k / k! lp[n, k], {k, 0, Log2@n}]
dlz[n_, z_] := lz[n, z] - lz[n - 1, z]
l2[n_, k_] := Sum[ (-1)^(k - j) Binomial[k, j] lz[n, j], {j, 0, k}]
dl2[n_, k_] := l2[n, k] - l2[n - 1, k]
l2z[n_, z_] := Expand@Sum[ bin[z, j] l2[n, j], {j, 0, Log2@n}]
FI[n_] := FactorInteger[n]; FI[1] := {}
dz[n_, z_] := Product[(-1)^p[[2]] Binomial[-z, p[[2]]], {p, FI[n]}]
dzx[n_, z2_] := Limit[Product[(p[[2]] + z - 1)! / ((z - 1)! p[[2]]!), {p, FI[n]}], z → z2]
dzy[n_, z_] := Product[Pochhammer[z, p[[2]]] / ( p[[2]]!), {p, FI[n]}]
ddz[n_, z_] := Product[z^p[[2]] / (p[[2]]!), {p, FI[n]}]
```

```
lz[100, z]
```

$$1 + 25\,z + 32\,z^2 + \frac{77\,z^3}{6} + \frac{35\,z^4}{12} + \frac{7\,z^5}{40} + \frac{7\,z^6}{720}$$

```
Table[dlz[n, 1], {n, 0, 32}]
```

$$\left\{0,\ 1,\ 1,\ 1,\ \frac{1}{2},\ 1,\ 1,\ 1,\ \frac{1}{6},\ \frac{1}{2},\ 1,\ 1,\ \frac{1}{2},\ 1,\ 1,\right.$$
$$\left.1,\ \frac{1}{24},\ 1,\ \frac{1}{2},\ 1,\ \frac{1}{2},\ 1,\ 1,\ 1,\ \frac{1}{6},\ \frac{1}{2},\ 1,\ \frac{1}{6},\ \frac{1}{2},\ 1,\ 1,\ 1,\ \frac{1}{120}\right\}$$

```
l2z[100, z]
```

$$1 + 25\,z + 32\,z^2 + \frac{77\,z^3}{6} + \frac{35\,z^4}{12} + \frac{7\,z^5}{40} + \frac{7\,z^6}{720}$$

```
dlz[7!, z]
```

$$\frac{z^8}{48}$$

```
ddz[7!, z]
```

$$\frac{z^8}{48}$$

```
Clear[n, z]
{dz[n = 160, z3 = -22 + .3 I], dzx[n, z3], dzy[n, z3]}
```

$$\{577438. - 51532.7\,i,\ 577438. - 51532.7\,i,\ 577438. - 51532.7\,i\}$$