

```

ClearAll["Global`*"]

WheelEntries := WheelEntries = 5
WheelSize := WheelSize = Product[Prime[j], {j, 1, WheelEntries}]
ExcludedPrimes[n_] := Sum[1/k, {j, 1, WheelEntries}, {k, 1, Log[Prime[j], n]}]

(*Use[n] is 1 if n isn't rejected by the wheel, and 0 otherwise*)
CoprimeCache := CoprimeCache = Table[CoprimeQ[WheelSize, n], {n, 1, WheelSize}]
Use[n_] := Use[n] = If[CoprimeCache[Mod[n - 1, WheelSize] + 1] == True, 1, 0]

(*Coprimes[n] is the count of numbers ≤ n coprime to our wheel*)
LegendrePhi[x_, a_] := LegendrePhi[x, a - 1] - LegendrePhi[x / Prime[a], a - 1];
LegendrePhi[x_, 0] := Floor[x]
LegPhiCache := LegPhiCache = Table[LegendrePhi[n, WheelEntries], {n, 1, WheelSize}]
FullWheel := FullWheel = LegendrePhi[WheelSize, WheelEntries]
Coprimes[n_] :=
  Coprimes[n] = LegPhiCache[Mod[n - 1, WheelSize] + 1] + Floor[(n - 1) / WheelSize] FullWheel

Dhyp[n_, k_, s_] := Sum[If[Use[m] == 0, 0, Binomial[k, j] Dhyp[Floor[n / (m^(k - j))], j, m + 1]],
  {m, s, n^(1/k)}, {j, 0, k - 1}]
Dhyp[n_, 1, s_] := Dhyp[n, 1, s] = Coprimes[n] - Coprimes[s - 1]
Dhyp[n_, 0, s_] := Dhyp[n, 0, s] = 1

d2cache[n_, k_] := Dhyp[n, k, 2] - Dhyp[n - 1, k, 2]
D2Cache[n_, k_] := Dhyp[n, k, 2]

D2Fast[n_, k_] :=
  Sum[If[Use[j] == 0, 0, d2cache[j, 1] D2Cache[Floor[n / j], k - 1]],
    {j, Floor[n^(1/3)] + 1, Floor[n^(1/2)]}] +
  Sum[(D2Cache[Floor[n / r], 1] - D2Cache[Floor[n / (r + 1)], 1]) D2Cache[r, k - 1],
    {r, 1, n / Floor[n^(1/2)] - 1}] +
  Sum[If[Use[j] == 0, 0, d2cache[j, k - 1] D2Cache[Floor[n / j], 1]], {j, 2, n^(1/3)}] +
  Sum[If[Use[j] == 0, 0, d2cache[j, m] D2Cache[Floor[n / (j s)], k - m - 1]],
    {j, 2, Floor[n^(1/3)]},
    {s, Floor[Floor[n^(1/3)] / j] + 1, Floor[Floor[n / j]^(1/2)]}, {m, 1, k - 2}] +
  Sum[If[Use[j] == 0 || Use[s] == 0, 0,
    (D2Cache[Floor[n / (j s)], 1] - D2Cache[Floor[n / (j (s + 1))], 1])
    (Sum[d2cache[j, m] D2Cache[s, k - m - 1], {m, 1, k - 2}]),
    {j, 2, Floor[n^(1/3)]}, {s, 1, Floor[n / j] / Floor[Floor[n / j]^(1/2)] - 1}]

D2Fast[n_, 1] := Coprimes[Floor[n]] - 1

LinnikSumFast[n_] := Sum[(-1)^(k + 1) / k D2Fast[n, k], {k, 1, Log[2, n]}]

RiePrimeCnt[n_] := Sum[PrimePi[n^(1/j)] / j, {j, 1, Log[2, n]}]
Table[{n, a = LinnikSumFast[n] + ExcludedPrimes[n], b = RiePrimeCnt[n], a - b},
  {n, 100, 3000, 100}] // TableForm

```

100	<u>428</u>	<u>428</u>	0
	15	15	
200	<u>5356</u>	<u>5356</u>	0
	105	105	
300	<u>56 981</u>	<u>56 981</u>	0
	840	840	
400	<u>23 707</u>	<u>23 707</u>	0
	280	280	
500	<u>28 467</u>	<u>28 467</u>	0
	280	280	
600	<u>293 023</u>	<u>293 023</u>	0
	2520	2520	
700	<u>333 973</u>	<u>333 973</u>	0
	2520	2520	
800	<u>369 673</u>	<u>369 673</u>	0
	2520	2520	
900	<u>408 733</u>	<u>408 733</u>	0
	2520	2520	
1000	<u>445 273</u>	<u>445 273</u>	0
	2520	2520	
1100	<u>97 169</u>	<u>97 169</u>	0
	504	504	
1200	<u>103 217</u>	<u>103 217</u>	0
	504	504	
1300	<u>110 777</u>	<u>110 777</u>	0
	504	504	
1400	<u>116 741</u>	<u>116 741</u>	0
	504	504	
1500	<u>125 309</u>	<u>125 309</u>	0
	504	504	
1600	<u>131 357</u>	<u>131 357</u>	0
	504	504	
1700	<u>139 169</u>	<u>139 169</u>	0
	504	504	
1800	<u>145 217</u>	<u>145 217</u>	0
	504	504	
1900	<u>151 517</u>	<u>151 517</u>	0
	504	504	
2000	<u>158 069</u>	<u>158 069</u>	0
	504	504	
2100	<u>1 816 879</u>	<u>1 816 879</u>	0
	5544	5544	
2200	<u>1 983 991</u>	<u>1 874 959</u>	<u>59</u>
	5544	5544	3
2300	<u>2 077 315</u>	<u>1 960 891</u>	<u>21</u>
	5544	5544	
2400	<u>2 167 867</u>	<u>2 044 051</u>	<u>67</u>
	5544	5544	3
2500	<u>2 233 933</u>	<u>2 100 877</u>	<u>24</u>
	5544	5544	
2600	<u>2 298 613</u>	<u>2 161 861</u>	<u>74</u>
	5544	5544	3
2700	<u>2 394 709</u>	<u>2 245 021</u>	<u>27</u>
	5544	5544	
2800	<u>2 481 565</u>	<u>2 322 637</u>	<u>86</u>
	5544	5544	3
2900	<u>2 558 257</u>	<u>2 391 937</u>	<u>30</u>
	5544	5544	
3000	<u>2 626 633</u>	<u>2 452 921</u>	<u>94</u>
	5544	5544	3