

```

DD[A_, k_, n_] := Sum[j^A DD[A, k - 1, n / j], {j, 2, n}]
DD[A_, 1, n_] := Sum[j^A, {j, 2, n}]
d[A_, k_, n_] := DDFast[A, k, n] - DDFast[A, k, n - 1]
rng[0, start_, end_] := Floor[end] - (start - 1)
rng[1, start_, end_] := Floor[end] (Floor[end] + 1) / 2 - (start - 1) start / 2
rng[2, start_, end_] :=
  Floor[end] (Floor[end] + 1) (2 Floor[end] + 1) / 6 - (start - 1) start (2 start - 1) / 6
rng[A_, start_, end_] := Sum[m^A, {m, start, end}]
DDFast[A_, 1, n_] := rng[A, 2, n]

DDFast[A_, k_, n_] := Sum[j^A DDFast[A, k - 1, n / j], {j, Floor[n^(1/3)] + 1, n^(1/2)}] +
  Sum[rng[A, Floor[n / (j + 1)] + 1, n / j] DDFast[A, k - 1, j], {j, 1, n / Floor[n^(1/2)] - 1}] +
  Sum[d[A, k - 1, j] rng[A, 2, n / j], {j, 2, n^(1/3)}] +
  Sum[s^A d[A, m, j] DDFast[A, k - m - 1, n / (j s)], {j, 2, n^(1/3)},
    {s, Floor[Floor[n^(1/3)] / j] + 1, Floor[n / j]^(1/2)}, {m, 1, k - 2}] +
  Sum[(rng[A, Floor[n / (j (s + 1))] + 1, n / (j s)])
    (Sum[d[A, m, j] DDFast[A, k - m - 1, s], {m, 1, k - 2}]),
    {j, 2, n^(1/3)}, {s, 1, Floor[n / j] / Floor[Floor[n / j]^(1/2)] - 1}]

SumPrimesFast[A_, n_] := Sum[(-1)^(k + 1) / (j k) MoebiusMu[j] DDFast[j A, k, n^(1/j)],
  {j, 1, Log[2, n]}, {k, 1, Log[2, (n^(1/j))]}]

SumPrimesFast[0, 1000]

```