

```

TestPrimePowerCount[A_, n_] := FullSimplify[Sum[MangoldtLambda[j] / Log[j] j^A, {j, 2, n}]]
ReferenceSumPrimes[A_, n_] :=
  Sum[1 / (j) MoebiusMu[j] TestPrimePowerCount[j A, n^(1 / j)], {j, 1, Log[2, n]}]
StrictDivisors[A_, k_, n_] := Sum[j^A StrictDivisors[A, k - 1, n / j], {j, 2, n}]
StrictDivisors[A_, 1, n_] := Sum[j^A, {j, 2, n}]
SumPrimes[A_, n_] := Sum[(-1)^(k + 1) / (j k) MoebiusMu[j] StrictDivisors[j A, k, n^(1 / j)],
  {j, 1, Log[2, n]}, {k, 1, Log[2, (n^(1 / j))]}]
RecurseCount[A_, k_, n_] := Sum[j^A (1 / k - RecurseCount[A, k + 1, n / j]), {j, 2, n}]
SumPrimesRecurse[A_, n_] :=
  Sum[1 / j MoebiusMu[j] RecurseCount[j A, 1, n^(1 / j)], {j, 1, Log[2, n]}]
StrictDivisorsHyperbola[A_, k_, n_, s_] :=
  Sum[(m^A)^(k - j) Binomial[k, j] StrictDivisorsHyperbola[A, j, n / (m^(k - j)), m + 1],
  {m, s, n^(1 / k)}, {j, 0, k - 1}]
StrictDivisorsHyperbola[A_, 1, n_, s_] := Sum[j^A, {j, s, n}]
StrictDivisorsHyperbola[0, 1, n_, s_] := Floor[n] - s + 1
StrictDivisorsHyperbola[1, 1, n_, s_] := Floor[n] (Floor[n] + 1) / 2 - s (s - 1) / 2
StrictDivisorsHyperbola[2, 1, n_, s_] :=
  Floor[n] (Floor[n] + 1) (2 Floor[n] + 1) / 6 - (s - 1) s (2 s - 1) / 6
StrictDivisorsHyperbola[3, 1, n_, s_] := Floor[n]^2 (Floor[n] + 1)^2 / 4 - s^2 (s - 1)^2 / 4
StrictDivisorsHyperbola[A_, 0, n_, s_] := 1
SumPrimesHyperbola[A_, n_] :=
  Sum[(-1)^(k + 1) / (j k) MoebiusMu[j] StrictDivisorsHyperbola[j A, k, n^(1 / j), 2],
  {j, 1, Log[2, n]}, {k, 1, Log[2, (n^(1 / j))]}]
Smallld[A_, k_, n_] := StrictDivisorsHyperbola[A, k, n, 2] -
  StrictDivisorsHyperbola[A, k, n - 1, 2]
StrictDivisorsReduced[a_, A_, k_, n_] :=
  Sum[Smallld[A, 1, j] StrictDivisors[A, k - 1, n / j], {j, a + 1, n}] +
  Sum[Smallld[A, k - 1, j] StrictDivisors[A, 1, n / j], {j, 2, a}] +
  Sum[Smallld[A, 1, s] Smallld[A, m, j] StrictDivisors[A, k - m - 1, n / (j s)],
  {j, 2, a}, {s, Floor[a / j] + 1, n / j}, {m, 1, k - 2}]
StrictDivisorsReduced[a_, A_, 1, n_] := Sum[j^A, {j, 2, n}]
SumPrimesReduced[A_, n_] := Sum[
  (-1)^(k + 1) / (j k) MoebiusMu[j] StrictDivisorsReduced[Floor[n^(1 / 3)], j A, k, n^(1 / j)],
  {j, 1, Log[2, n]}, {k, 1, Log[2, (n^(1 / j))]}]
StrictDivisorsFullReduced[A_, k_, n_] :=
  Sum[j^A StrictDivisorsHyperbola[A, k - 1, n / j, 2], {j, Floor[n^(1 / 3)] + 1, n^(1 / 2)}] +
  Sum[Sum[m^A, {m, Floor[n / (j + 1)] + 1, n / j}] StrictDivisorsHyperbola[A, k - 1, j, 2],
  {j, 1, n / Floor[n^(1 / 2)] - 1}] +
  Sum[Smallld[A, k - 1, j] Sum[m^A, {m, 2, n / j}], {j, 2, n^(1 / 3)}] +
  Sum[s^A Smallld[A, m, j] StrictDivisorsHyperbola[A, k - m - 1, n / (j s), 2],
  {j, 2, n^(1 / 3)}, {s, Floor[Floor[n^(1 / 3)] / j] + 1, Floor[n / j]^(1 / 2)}, {m, 1, k - 2}] +
  Sum[(Sum[m^A, {m, Floor[n / (j (s + 1))] + 1, n / (j s)}])
  (Sum[Smallld[A, m, j] StrictDivisorsHyperbola[A, k - m - 1, s, 2], {m, 1, k - 2}]),
  {j, 2, n^(1 / 3)}, {s, 1, Floor[n / j] / Floor[Floor[n / j]^(1 / 2)] - 1}]
StrictDivisorsFullReduced[A_, 1, n_] := Sum[j^A, {j, 2, n}]
SumPrimesFullReduced[A_, n_] :=
  Sum[(-1)^(k + 1) / (j k) MoebiusMu[j] StrictDivisorsFullReduced[j A, k, n^(1 / j)],
  {j, 1, Log[2, n]}, {k, 1, Log[2, (n^(1 / j))]}]

```

```
SumPrimesFullReduced[1, 100]
```

```
1060
```