5. Zeroes of Exponential-Style Dirichlet Convolutions

Starting with $[\zeta(s)^z]_n$ and using n a fixed value and z a complex variable, this section explores the roots of $[\zeta(s)^z]_n$ and shows how they can be used in expressions for $[\zeta(s)^z]_n$, $\Pi(n)$, and the Mertens function. It then presents several other similar sets of zeros, including one for the Chebyshev function $\psi(n)$.

5.1 A Worked Example: The Zeros of $[\zeta(s)^z]_n$ for Fixed n and s

Given we can define the divisor sum function for complex z $[\zeta(s)^z]_n$ using $[(\log\zeta(s))^k]_n$ (from (1.6)) as

$$[(\log\zeta(s))^k]_n = \sum_{j=2} j^{-s} \frac{\Lambda(j)}{\log j} [(\log\zeta(s))^{k-1}]_{n j^{-1}} \text{ and } [\zeta(s)^z]_n = \sum_{k=0}^{\infty} \frac{z^k}{k!} [(\log\zeta(s))^k]_n$$

```
logzeta[n_,s_,0]:=UnitStep[n-1]
logzeta[n_,s_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] logzeta[n/j,s,k-1],{j,2,n}]
zeta[n_,s_,z_]:=Sum[z^k/k! logzeta[n,s,k],{k,0,Log[2,n]}]
```

and given that $[(\log\zeta(s))^k]_n = 0$ when $n < 2^k$, then for some fixed n, $[\zeta(s)^z]_n$ can be treated as a polynomial of degree $\log_2 n$ with z as its variable. As an example,

$$\begin{aligned}[\zeta(0)^z]_{100} &= \sum_{k=0}^{\lfloor \log_2 100 \rfloor} \frac{z^k}{k!} [(\log\zeta(0))^k]_{100} = \\ &1 + \frac{428}{15} z + \frac{16289}{360} z^2 + \frac{331}{16} z^3 + \frac{611}{144} z^4 + \frac{67}{240} z^5 + \frac{7}{720} z^6\end{aligned} \tag{5.1.1}$$

```
logzeta[n_,s_,0]:=UnitStep[n-1]
logzeta[n_,s_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] logzeta[n/j,s,k-1],{j,2,n}]
zeta[n_,s_,z_]:=Sum[z^k/k! logzeta[n,s,k],{k,0,Log[2,n]}]
zeta[100,0,z]
```

Thus, it should have $\log_2 n$ solutions for z where $[\zeta(0)^z]_n = 0$.

```
logzeta[n_,s_,0]:=UnitStep[n-1]
logzeta[n_,s_,k_]:=Sum[FullSimplify[MangoldtLambda[j]/Log[j]] logzeta[n/j,s,k-1],{j,2,n}]
zeta[n_,s_,z_]:=Sum[z^k/k! logzeta[n,s,k],{k,0,Log[2,n]}]
Table[{n,Roots[ zeta[n,0,z]==0,z]},{n,2,31}]//TableForm
```

5.2 The General Form of Convolution Zeros

Suppose we have some partial sum convolution $[f^k]_n$ where $k \in \mathbb{N}$. And suppose there is some j for which $[f^k]_n = 0$ when k > j.

Given the convolution

$$[(1+f)^z]_n = \sum_{k=0}^{j} \binom{z}{k} [f^k]_n \tag{5.2.1}$$

there will be *j* values of *z* such that

$$[(1+f)^z]_n = 0 \tag{5.2.2}$$

If we denote those values as ρ, then we can express $[(1+f)^z]_n$ in terms of these zeros as

$$[(1+f)^z]_n = \prod_{\rho} \left(1 - \frac{z}{\rho}\right) \tag{5.2.3}$$

If we further have $[\log(1+f)]_n = \lim_{z \to 0} \frac{[(1+f)^z]_n - 1}{z}$ then

$$[\log(1+f)]_n = -\sum_{\rho} \frac{1}{\rho} \tag{5.2.4}$$

5.3 $[\zeta(s)^z]_n$ as a Product of its Zeros

Denote the roots ρ, and, through a bit of algebraic manipulation, and because $[\zeta(s)^0]_n = 1$, we have

$$\boxed{[\zeta(s)^z]_n = \prod_{\rho} \left(1 - \frac{z}{\rho}\right)} \tag{5.3.1}$$

```
ri[]:=RandomInteger[{10,100}];rr[]:=RandomReal[{-3,3}]+RandomReal[{-3,3}]I
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
zeros[n_,s_] :=List@@Roots[ zeta[n,s,z,1]==0,z][[All,2]]
zetaalt[n_, s_,z_] := Product[ 1-z/r,{r,zeros[n,s]}]
Table[Chop[zeta[a=ri[],b=rr[],c=rr[],1]-zetaalt[a,b,c]],{j,1,100}]
```

5.4 $\Pi(n)$ as a Sum of the Zeros of $[\zeta(0)^z]_n$

$$\boxed{[\log \zeta(s)]_n = -\sum_{\rho} \frac{1}{\rho}}$$

These zeros are connected to $\Pi(n)$, the Riemann Prime counting function, as

$$\boxed{\Pi(n) = -\sum_{\rho} \frac{1}{\rho}}$$

(5.4.1)

```
rr[]:=RandomReal[{-3,3}]+RandomReal[{-3,3}]I
logzeta[n_,s_,0]:=UnitStep[n-1]
logzeta[n_,s_,k_]:=Sum[j^-s FullSimplify[MangoldtLambda[j]/Log[j]] logzeta[n/j,s,k-1],{j,2,n}]
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
zeros[n_,s_] :=List@@Roots[ zeta[n,s,z,1]==0,z][[All,2]]
logzetaalt[n_,s_] := FullSimplify[-Sum[ 1/r,{r,zeros[n,s]}]]
Table[Chop[logzeta[n,a=rr[],1]-logzetaalt[n,a]],{n,4,25}]
```

5.5 Other Noteworthy Products of the Zeros of $[\zeta(0)^z]_n$

Here are specific results with these roots, with $M(n)$ Mertens function ($[\zeta(0)^{-1}]_n$ in this paper) and $D(n)$ the standard Dirichlet Divisor problem ($[\zeta(0)^2]_n$ in this paper)

$$M(n) = \prod_{\rho} \left(1 + \frac{1}{\rho}\right) \qquad 1 = \prod_{\rho} \left(1 - \frac{0}{\rho}\right)$$
$$\lfloor n \rfloor = \prod_{\rho} \left(1 - \frac{1}{\rho}\right) \qquad D(n) = \prod_{\rho} \left(1 - \frac{2}{\rho}\right)$$

(5.5.1)

```
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
zeros[n_,s_]:=List@@NRoots[zeta[n,s,z,1]==0,z][[All,2]]
prod[n_,s_,z_]:=Product[1-z/r,{r,zeros[n,s]}]
Table[{Chop[Sum[MoebiusMu[j],{j,1,n}]-prod[n,0,-1]],Chop[1-prod[n,0,0]],Chop[n-prod[n,0,1]],Chop[Sum[1,{j,1,n},
{k,1,Floor[n/j]}]-prod[n,0,2]]},{n,4,100}]
```

5.6 The Zeros of $\frac{[\zeta(s)^z]_n - 1}{z}$

A close variant of this idea is

$$\frac{[\zeta(s)^z]_n - 1}{z} = \sum_{k=1}^{\infty} \frac{z^{k-1}}{k!} [(\log \zeta(s))^k]_n$$

(5.6.1)

```
logzeta[n_,s_,0]:=UnitStep[n-1]
logzeta[n_,s_,k_]:=Sum[j^-s MangoldtLambda[j]/Log[j] logzeta[n/j,k-1],{j,2,Floor[n]}]
zetam1z[n_,s_,z_]:=Sum[z^(k-1)/k! logzeta[n,s,k],{k,1,Log[2,n]}]
```

This function, for a fixed n, has $\log_2 n - 1$ roots, denoted ρ. With these roots, $[\zeta(s)^z]_n$ is

$$\boxed{[\zeta(s)^z]_n = 1 + z \cdot [\log \zeta(s)]_n \cdot \prod_{\rho} \left(1 - \frac{z}{\rho}\right)}$$

(5.6.2)

```
ri[]:=RandomInteger[{10,100}];rr[]:=RandomReal[{-3,3}]+RandomReal[{-3,3}]I
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
zeros[n_,s_]:=List@@NRoots[(zeta[n,s,z,1]-1)/z==0,z][[All,2]]
zetaalt[n_,s_,z_]:=1+z Limit[D[zeta[n,s,y,1],y],y->0]Product[1-z/r,{r,zeros[n,s]}]
Table[Chop[zeta[a=ri[],b=rr[],c=rr[],1]-zetaalt[a,b,c]],{j,1,100}]
```

More results with these zeros include

$$M(n)=1-\Pi(n)\cdot\prod_{\rho}(1+\frac{1}{\rho})$$
$$\Pi(n)=\Pi(n)\cdot\prod_{\rho}(1-\frac{0}{\rho})$$
$$\lfloor n\rfloor=1+\Pi(n)\cdot\prod_{\rho}(1-\frac{1}{\rho})$$
$$D(n)=1+2\,\Pi(n)\cdot\prod_{\rho}(1-\frac{2}{\rho})$$

(5.6.6)

```
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
zeros[n_,s_]:=List@@NRoots[(zeta[n,s,z,1]-1)/z==0,z][[All,2]]
prod[n_,s_,z_]:=1+z Limit[D[ zeta[n,s,y,1],y],y->0]Product[1-z/r,{r,zeros[n,0]}]
Table[{Chop[Sum[MoebiusMu[j],{j,1,n}]-prod[n,0,-1]],Chop[n-prod[n,0,1]],Chop[Sum[1,{j,1,n},{k,1,Floor[n/j]}]-prod[n,0,2]]},{n,8,100}]
```

5.7 The Zeros of $[(1+\log\zeta(s))^z]_n$

Here's another interesting set of zeros. Starting with $[(\log\zeta(s))^k]_n=\sum_{j=2} j^{-s}\frac{\Lambda(j)}{\log j}[(\log\zeta(s))^{k-1}]_{n\,j^{-1}}$ from (1.6), we define $[(1+\log\zeta(s))^z]_n$, a prime power analog to $[\zeta(s)^z]_n$, for which $[1+\log\zeta(0)]_n=\Pi(n)+1$:

$$[(1+\log\zeta(s))^z]_n=\sum_{k=0}^{\infty}\binom{z}{k}[(\log\zeta(s))^k]_n$$

(5.7.1)

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
logzeta[n_,s_,k_]:=Limit[D[zeta[n,s,z,1],{z,k}],z->0]
logzetaplus1[ x_, s_, z_ ] := Sum[ bin[ z, k ] logzeta[ x, s, k ], { k, 0, Log[ 2, x ] } ]
```

Although less obvious than the above cases, (4.6) also gives, for fixed *n,* a polynomial of degree $\log_2 n$, using the complex generalized binomial coefficient $\binom{z}{k}=\frac{z(z-1)...(z-k+1)}{k!}$. For example,

$$[(1+\log\zeta(0))^z]_{12}=\sum_{k=0}^{\lfloor\log_2 12\rfloor}\binom{z}{k}[(\log\zeta(0))^k]_{12}=$$
$$(1)+\frac{z}{1!}(\frac{19}{3})+\frac{z\cdot(z-1)}{2!}(8)+\frac{z\cdot(z-1)\cdot(z-2)}{3!}(4)$$

$$[(1+\log\zeta(0))^z]_{12}=1+\frac{11}{3}z+2z^2+\frac{2}{3}z^3$$

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
zeta[n_,s_,z_,k_]:=1+((z+1)/k-1)Sum[j^-s zeta[n/j,s,z,k+1],{j,2,n}]
logzeta[n_,s_,k_]:=Limit[D[zeta[n,s,z,1],{z,k}],z->0]
logzetaplus1[ x_, s_, z_ ] := Sum[ bin[ z, k ] logzeta[ x, s, k ], { k, 0, Log[ 2, x ] } ]
Expand[logzetaplus1[12,0,z]]
```

Like for $[\zeta(s)^z]_n$ above, $[(1+\log\zeta(s))^z]_n$, for fixed *n*, has $\log_2 n$ values of *z* for which $[(1+\log\zeta(s))^z]_n=0$, denoted ρ . If s is 0, with those roots, we have another Riemann Prime counting function identity:

$$\Pi(n) = -1 + \prod_{\rho} \left(1 - \frac{1}{\rho}\right) \tag{5.7.2}$$

```
bin[ z_, k_ ] := Product[ z-j, { j, 0, k-1 } ]/k!
RiemannPrimeCount[n_]:=Sum[PrimePi[n^(1/j)]/j,{j,1,Log[2,n]}]
logD[n_,k_]:=Limit[D[Dz[n,z,1],{z,k}],z->0];Dz[n_,z_,k_]:=1+((z+1)/k-1)Sum[Dz[n/j,z,k+1],{j,2,n}]
logDplus1[ x_, z_ ] := Sum[ bin[ z, k ] logD[ x, k ], { k, 0, Log[ 2, x ] } ]
zeros[n_] :=List@@NRoots[ logDplus1[n,z]==0,z][[All,2]]
Table[Chop[N[RiemannPrimeCount[n]]-(-1+Product[ 1-1/r,{r,zeros[n]}])],{n,4,100}]//TableForm
```